 #-------------------------------------------------
#
# Project created by QtCreator 2011-10-18T19:39:10
#
#-------------------------------------------------

QT += core gui network xml
TARGET = DTD
TEMPLATE = app

#Add fmod to CONFIG, then re-qmake/rebuild for sound using FMOD
CONFIG += fmod

INCLUDEPATH *= $$PWD/backend \
   $$PWD/frontend \
   $$PWD/../deps/fmod/inc

fmod {
   DEFINES += USE_FMOD
   win32-g++ {
      LIBS += $$PWD/../deps/fmod/lib/libfmodex.a
   } else:win32 {
      LIBS += $$PWD/../deps/fmod/lib/fmodex_vc.lib
   }
}

SOURCES += \
    frontend/main.cpp \
    frontend/mainwindow.cpp \
    frontend/newgamedialog.cpp \
    frontend/gameview.cpp \
    frontend/client.cpp \
    backend/game.cpp \
    backend/unit.cpp \
    backend/server.cpp \
    backend/player.cpp \
    frontend/clientgame.cpp \
    backend/utilities.cpp \
    backend/tower.cpp \
    backend/entity.cpp \
    backend/entitymanager.cpp \
    frontend/cliententity.cpp \
    frontend/colorwheel.cpp \
    frontend/soundsystem.cpp \
    frontend/entitypreview.cpp \
    frontend/inventorybutton.cpp \
    frontend/inventory.cpp \
    backend/unitemitter.cpp \
    frontend/gfxitems/gfxUnitEmitter.cpp \
    frontend/gfxitems/gfxTowerBeam.cpp \
    frontend/gfxitems/gfxUnit.cpp \
    frontend/gfxitems/gfxItem.cpp \
    frontend/gfxitems/gfxDamageTextFX.cpp \
    frontend/gfxitems/gfxUnitDeathFX.cpp \
    frontend/gfxitems/gfxTowerBeamFireFX.cpp
fmod {
   SOURCES += frontend/soundsystemfmod.cpp
}

HEADERS  +=  \
    frontend/mainwindow.h \
    frontend/newgamedialog.h \
    frontend/gameview.h \
    frontend/client.h \
    backend/game.h \
    backend/unit.h \
    backend/server.h \
    backend/player.h \
    frontend/clientgame.h \
    backend/utilities.h \
    frontend/render.h \
    backend/tower.h \
    backend/entity.h \
    backend/entitymanager.h \
    frontend/cliententity.h \
    frontend/gfxitems/unitdeathfx.h \
    frontend/colorwheel.h \
    frontend/soundsystem.h \
    frontend/entitypreview.h \
    frontend/inventorybutton.h \
    frontend/inventory.h \
    backend/unitemitter.h \
    frontend/gfxitems/gfxUnitEmitter.h \
    frontend/gfxitems/gfxTowerBeam.h \
    frontend/gfxitems/gfxUnit.h \
    frontend/gfxitems/gfxItem.h \
    frontend/gfxitems/gfxDamageTextFX.h \
    frontend/gfxitems/gfxUnitDeathFX.h \
    frontend/gfxitems/gfxTowerBeamFireFX.h
fmod {
   HEADERS += frontend/soundsystemfmod.h
}

FORMS    += \
    frontend/mainwindow.ui \
    frontend/newgamedialog.ui

CONFIG += console

RESOURCES += \
    ../exe/res/assets.qrc

message($$QMAKESPEC)
